# Solving a Package problem

```
> restart:
  with(LinearAlgebra):
```

Function to minimize

```
> f := -x*y*z ;
```

$$f := -x\,y\,z \quad (1)$$

4 constraints

```
> h  := 2*(x*y+y*z+x*z)-1 ;
```
Surface constraint
```
  g1 := x ;
  g2 := y ;
  g3 := z ;
```

$$h := 2\,x\,y + 2\,y\,z + 2\,x\,z - 1$$
$$g1 := x$$
$$g2 := y$$
$$g3 := z \quad (2)$$

Build the Lagrangian

```
> L := f - lambda*h - mu1*g1 - mu2*g2 - mu3*g3 ;
```

$$L := -x\,y\,z - \lambda\,(2\,x\,y + 2\,y\,z + 2\,x\,z - 1) - \mu 1\,x - \mu 2\,y - \mu 3\,z \quad (3)$$

Build the nonlinear system with KKT conditions

```
> EQ1 := diff(L,x) ;
  EQ2 := diff(L,y) ;
  EQ3 := diff(L,z) ;
  EQ4 := h ;

  EQ5 := g1 * mu1 ;
  EQ6 := g2 * mu2 ;
  EQ7 := g3 * mu3 ;
```

$$EQ1 := -y\,z - \lambda\,(2\,y + 2\,z) - \mu 1$$
$$EQ2 := -x\,z - \lambda\,(2\,x + 2\,z) - \mu 2$$
$$EQ3 := -x\,y - \lambda\,(2\,y + 2\,x) - \mu 3$$
$$EQ4 := 2\,x\,y + 2\,y\,z + 2\,x\,z - 1$$
$$EQ5 := \mu 1\,x$$
$$EQ6 := \mu 2\,y$$
$$EQ7 := \mu 3\,z \quad (4)$$

Solve the nonlinear system and discard solution with mu <0.

```
> solve( {EQ||(1..7)}, {x,y,z,lambda,mu1,mu2,mu3} ) : SOL :=
  allvalues([%]) ;
```

$$SOL := \left[\left\{\lambda = 0, \mu 1 = -\frac{1}{2}, \mu 2 = 0, \mu 3 = 0, x = 0, y = y, z = \frac{1}{2\,y}\right\}, \left\{\lambda = 0, \mu 1 = 0, \mu 2 = -\frac{1}{2}, \right. \quad (5)$$

$$\mu 3 = 0, x = x, y = 0, z = \frac{1}{2\,x}\right\}, \left\{\lambda = 0, \mu 1 = 0, \mu 2 = 0, \mu 3 = -\frac{1}{2}, x = x, y = \frac{1}{2\,x}, z = 0\right\}, \left\{\lambda\right.$$

$= -\frac{1}{24}\sqrt{6}, \mu 1 = 0, \mu 2 = 0, \mu 3 = 0, x = \frac{1}{6}\sqrt{6}, y = \frac{1}{6}\sqrt{6}, z = \frac{1}{6}\sqrt{6}\}\Big], \Big[\{\lambda = 0, \mu 1 = -\frac{1}{2}, \mu 2 = 0, \mu 3 = 0, x = 0, y = y, z = \frac{1}{2y}\}, \{\lambda = 0, \mu 1 = 0, \mu 2 = -\frac{1}{2}, \mu 3 = 0, x = x, y = 0, z = \frac{1}{2x}\}, \{\lambda = 0, \mu 1 = 0, \mu 2 = 0, \mu 3 = -\frac{1}{2}, x = x, y = \frac{1}{2x}, z = 0\}, \{\lambda = \frac{1}{24}\sqrt{6}, \mu 1 = 0, \mu 2 = 0, \mu 3 = 0, x = -\frac{1}{6}\sqrt{6}, y = -\frac{1}{6}\sqrt{6}, z = -\frac{1}{6}\sqrt{6}\}\Big]\Big]$

```
> SOL1 := SOL[1][4] ;
```

$$SOL1 := \{\lambda = -\frac{1}{24}\sqrt{6}, \mu 1 = 0, \mu 2 = 0, \mu 3 = 0, x = \frac{1}{6}\sqrt{6}, y = \frac{1}{6}\sqrt{6}, z = \frac{1}{6}\sqrt{6}\} \quad (6)$$

```
> SOL2 := SOL[2][4];
```

$$SOL2 := \{\lambda = \frac{1}{24}\sqrt{6}, \mu 1 = 0, \mu 2 = 0, \mu 3 = 0, x = -\frac{1}{6}\sqrt{6}, y = -\frac{1}{6}\sqrt{6}, z = -\frac{1}{6}\sqrt{6}\} \quad (7)$$

We found 2 solution:

Compute the hessian

```
> Hess := <<diff(L,x,x),diff(L,x,y),diff(L,x,z)>|
           <diff(L,y,x),diff(L,y,y),diff(L,y,z)>|
           <diff(L,z,x),diff(L,z,y),diff(L,z,z)>>;
```

$$Hess := \begin{bmatrix} 0 & -z-2\lambda & -y-2\lambda \\ -z-2\lambda & 0 & -x-2\lambda \\ -y-2\lambda & -x-2\lambda & 0 \end{bmatrix} \quad (8)$$

```
> Hess1 := subs(SOL1,Hess) ;
```

$$Hess1 := \begin{bmatrix} 0 & -\frac{1}{12}\sqrt{6} & -\frac{1}{12}\sqrt{6} \\ -\frac{1}{12}\sqrt{6} & 0 & -\frac{1}{12}\sqrt{6} \\ -\frac{1}{12}\sqrt{6} & -\frac{1}{12}\sqrt{6} & 0 \end{bmatrix} \quad (9)$$

g1, g2, g3 are not active

```
> H := <diff(h,x)|diff(h,y)|diff(h,z)> ;
```

$$H := \begin{bmatrix} 2y+2z & 2x+2z & 2y+2x \end{bmatrix} \quad (10)$$

```
> H1 := subs(SOL1,H) ;
```

$$H1 := \begin{bmatrix} \frac{2}{3}\sqrt{6} & \frac{2}{3}\sqrt{6} & \frac{2}{3}\sqrt{6} \end{bmatrix} \quad (11)$$

```
> Z1, Z2 := op(NullSpace(H1)) ;
```

$$Z1, Z2 := \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} -1 \\ 1 \\ 0 \end{bmatrix} \quad (12)$$

```
> HessReduced1 := Transpose(<Z1|Z2>).Hess1.<Z1|Z2> ;
```

$$HessReduced1 := \begin{bmatrix} \frac{1}{6}\sqrt{6} & \frac{1}{12}\sqrt{6} \\ \frac{1}{12}\sqrt{6} & \frac{1}{6}\sqrt{6} \end{bmatrix} \quad (13)$$

Using Sylvester is positive definite

```
> Determinant(<HessReduced1[1,1]>) ;
  Determinant(<HessReduced1[1..2,1..2]>) ;
```

$$\frac{1}{6}\sqrt{6}$$

$$\frac{1}{8} \quad (14)$$

```
> Hess2 := subs(SOL2,Hess) ;
```

$$Hess2 := \begin{bmatrix} 0 & \frac{1}{12}\sqrt{6} & \frac{1}{12}\sqrt{6} \\ \frac{1}{12}\sqrt{6} & 0 & \frac{1}{12}\sqrt{6} \\ \frac{1}{12}\sqrt{6} & \frac{1}{12}\sqrt{6} & 0 \end{bmatrix} \quad (15)$$

g1, g2, g3 are not active

```
> H2 := subs(SOL2,H) ;
```

$$H2 := \begin{bmatrix} -\frac{2}{3}\sqrt{6} & -\frac{2}{3}\sqrt{6} & -\frac{2}{3}\sqrt{6} \end{bmatrix} \quad (16)$$

```
> Z1, Z2 := op(NullSpace(H2)) ;
```

$$Z1, Z2 := \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} -1 \\ 1 \\ 0 \end{bmatrix} \quad (17)$$

```
> HessReduced2 := Transpose(<Z1|Z2>).Hess2.<Z1|Z2> ;
```

$$HessReduced2 := \begin{bmatrix} -\frac{1}{6}\sqrt{6} & -\frac{1}{12}\sqrt{6} \\ -\frac{1}{12}\sqrt{6} & -\frac{1}{6}\sqrt{6} \end{bmatrix} \quad (18)$$

Using Sylvester is positive definite

```
> Determinant(<HessReduced2[1,1]>) ;
  Determinant(<HessReduced2[1..2,1..2]>) ;
```

$$-\frac{1}{6}\sqrt{6}$$

$$\frac{1}{8} \quad (19)$$